  

	***********************************************************************************
	**																				 **
	**																				 **
	**		NAME:		ALLISON CARNEGIE & NIKHAR GAIKWAD 							 **
	**		DATE: 		January 7, 2022 											 **
	**		PROJECT: 	Public Opinion on Geopolitics and Trade: Theory and Evidence **
	**																				 **
	**		DETAILS: 	This file is written to be run to conduct the analysis 		 **
	**					of the India survey data 									 **
	**					Data drawn from file: India_SurveyData_insheet.csv			 **
	**					Stata version 14.2											 **
	**																				 **
	***********************************************************************************

set more off

cd ".../ReplicationArchive/Dataverse"
* Replace with file pathway to replication files on one's own computer

insheet using "Data/India_SurveyData_insheet.csv", clear

****************************************
* Conduct Vignette Experiment Analysis *
****************************************

*******************
* Main Manuscript *
*******************

************
* Figure 5 *
************

graph bar (mean) support_trade, ///
over(adversary_treat, relabel(1 "Ally" 2 "Adversary")) ///
bar(1, fcolor(gray) lcolor(none)) ///
ytitle("Mean Support for Trade") ///
ylabel(0(.2).8, nogrid noticks) ///
blabel(bar, format(%4.3f)) ///
graphregion(color(white)) bgcolor(white) scheme(s2mono) ///
title("{bf:Support for Trade by Partner's Geopolitical Status: India Sample}", size(4) color(black) span) ///
note("Difference = 0.180 (p = 0.000).  Difference and p-value are based on a two-sided t-test.  N=480.")
graph export "Output/Figure5.pdf", replace


************
* Figure 6 *
************

graph bar (mean) support_trade, over(military_treat, ///
relabel(1 "Not Help" 2 "Help Military")) ///
bar(1, fcolor(gray) lcolor(none)) ///
ytitle("Mean Support for Trade") ///
ylabel(0(.2).8, nogrid noticks) ///
blabel(bar, format(%4.3f)) ///
graphregion(color(white)) bgcolor(white) scheme(s2mono) ///
note("   Difference = 0.094 (p = 0.033)", justification(center) ) ///
name(figure1b,replace) 

graph bar (mean) support_trade, over(peace_treat, ///
relabel(1 "Not Foster" 2 "Foster Peace")) ///
bar(1, fcolor(gray) lcolor(none)) ///
ylabel(0(.2).8, nogrid noticks) ///
blabel(bar, format(%4.3f)) ///
graphregion(color(white)) bgcolor(white) scheme(s2mono) ///
note("   Difference = -0.217 (p = 0.000)", justification(center)) ///
name(figure1c,replace)  yscale(off fill)

graph bar (mean) support_trade, over(democracy_treat, ///
relabel(1 "Is Not" 2 "Is Democracy")) ///
bar(1, fcolor(gray) lcolor(none)) ///
ylabel(0(.2).8, nogrid noticks) ///
blabel(bar, format(%4.3f)) ///
graphregion(color(white)) bgcolor(white) scheme(s2mono) ///
note("  Difference = -0.027 (p = 0.541)", justification(center)) ///
name(figure1d,replace) yscale(off fill)

graph combine figure1b figure1c figure1d, ///
ycommon cols(3)  ///
graphregion(color(white)) scheme(s2mono) ///
title("{bf:Support for Trade by Randomized Attributes: India Sample}", size(4) color(black) span) ///
note("     Differences and p-values are based on two-sided t-tests.  N=480.", justification(center))

graph export "Output/Figure6.pdf", replace


**************************
* Supplementary Appendix *
**************************

************************
* Appendix G: Table 11 *
************************

* Column 1 (Ally/Adversary Treatment, with controls)

reg support_trade adversary_treat gender age educ religion caste income, robust

* Column 2 (Military Size Treatment, with controls)

reg support_trade military_treat gender age educ religion caste income, robust

* Column 3 (Peace Treatment, with controls)

reg support_trade peace_treat gender age educ religion caste income, robust

* Column 4 (Democracy Treatment, with controls)

reg support_trade democracy_treat gender age educ religion caste income, robust
